MODIFIED INTEGRATION QUADRATURES WITH MAPLE

Wei-Chi Yang
Radford University
Department of Math and Stat, Radford, VA 24142
e-mail: wyang@mathstat.ms.runet.edu

Myungsuk Chung
Virginia Polytechnic Institute and State University
Department of Mathematics, Blacksburg, VA 24061
e-mail: myungsuk@math.vt.edu

In this paper, we will demonstrate how nonadaptive integration quadratures such as Simpson's Rule, and Trapezoidal's Rule with equally spaced divisions can be easily modified into adapative quadratures with non-equally spaced divisions by using "matrices or arrays". The necessicity of adaptive quadratures depends upon the behavior of a function. These quadratures can be used in treating functions which are monotone, or highly oscillating with singularities (see [2]). All these quadratures could not have been experimented without the presence of computer algebra system such as Maple. We assume readers are familiar with basic Maple commands. The following example will show why the adapted quadratures are necessary and how we use matrices to device such quadratures:

Example: Let $f(x) = 1/\sqrt{x}$ for $x \in (0, 1]$, and $f(0) = 0$. We use Maple to demonstrate how we estimate the integral of f over the interval [0,1].

> *f := proc(x) 1/sqrt(x)end;*

f := proc(x) 1/sqrt(x) end

> *f(x);*

$$\frac{1}{\sqrt{x}}$$

Now we define the matrix "*ank*" which is used to determine the widths of each subinterval we choose.

> *ank:=proc(a,b,n,k) 2*(b-a)*k/(n*(n+1)) end;*

ank := proc(a,b,n,k) 2*(b-a)*k/n/(n+1) end

> *ank(a, b, n, k);*

$$2\frac{(b-a)\,k}{n\,(n+1)}$$

Notice that the sum of "ank" from $k = 1$ to n is $b - a$, this is because the matrix ank is formed by using the formula of $\sum_{k=1}^{n} k = \frac{n(n+1)}{2}$.

> $simplify(sum(ank(a, b, n, k), k = 1..n));$

$$b - a$$

Now let's print out the matrix formed by "ank", which will tell us how we divide our interval.

> $with(linalg):$

>*A:=matrix(9,9, proc(i,j) if j > i then 0 else ank(0,1,i,j) fi end);*

$$A := \begin{bmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ \frac{1}{3} & \frac{2}{3} & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ \frac{1}{6} & \frac{1}{3} & \frac{1}{2} & 0 & 0 & 0 & 0 & 0 & 0 \\ \frac{1}{10} & \frac{1}{5} & \frac{3}{10} & \frac{2}{5} & 0 & 0 & 0 & 0 & 0 \\ \frac{1}{15} & \frac{2}{15} & \frac{1}{5} & \frac{4}{15} & \frac{1}{3} & 0 & 0 & 0 & 0 \\ \frac{1}{21} & \frac{2}{21} & \frac{1}{7} & \frac{4}{21} & \frac{5}{21} & \frac{2}{7} & 0 & 0 & 0 \\ \frac{1}{28} & \frac{1}{14} & \frac{3}{28} & \frac{1}{7} & \frac{5}{28} & \frac{3}{14} & \frac{1}{4} & 0 & 0 \\ \frac{1}{36} & \frac{1}{18} & \frac{1}{12} & \frac{1}{9} & \frac{5}{36} & \frac{1}{6} & \frac{7}{36} & \frac{2}{9} & 0 \\ \frac{1}{45} & \frac{2}{45} & \frac{1}{15} & \frac{4}{45} & \frac{1}{9} & \frac{2}{15} & \frac{7}{45} & \frac{8}{45} & \frac{1}{5} \end{bmatrix}$$

Remark: (i) Notice that the 9 by 9 matrix A represents a partial list as to how we choose the length of each subinterval. For example if we divide the interval [0,1] into 9 subintervals, then the lengths of each subinterval are 1/45, 2/45, and so forth. Also note that the sum of each row is 1 and the (n, k)-entry $a(n, k)$ tends to 0 as n tends to infinity regardless of k, see [2, Definition 2].
(ii) We remark that the function f is steeper when x is close to 0, that is why we choose the sequence ank decreases to 0 or the entries of each row, n, of the matrix A satisfying $a(n, i) < a(n, j)$ if $i < j$. This suggests that when one considers the division of the definition of an integral, he or she should take the local behavior of the function into consideration. That is why we have generalized Riemann integral and Henstock integral, see [1], [3], or [4].
(iii) We can continue dividing the interval [0,1] into any n- subintervals by using the formula of ank. Now we shall describe our first quadrature. Intuitively, we are using trapezoidal's rule with the matrix determined by "ank". That is why we call it an adaptive trapezoidal rule.

First we define the right and left end evaluating points:

> *right := proc(a, b, j, k, n) a+sum(ank(a,b,n,j), j=1..k) end;*

right := proc(a,b,j,k,n) a+sum(ank(a,b,n,j),j = 1 .. k) end

> *right(a, b, j, k, n);*

$$a+\frac{b\,(k+1)^2}{n\,(n+1)}-\frac{b\,(k+1)}{n\,(n+1)}-\frac{a\,(k+1)^2}{n\,(n+1)}+\frac{a\,(k+1)}{n\,(n+1)}$$

> *left:=proc(a,b,j,k,n) a+sum(ank(a,b,n,j), j=0..k-1) end;*

left := proc(a,b,j,k,n) a+sum(ank(a,b,n,j),j = 0 .. k-1) end

>*left(a,b,j,k,n);*

$$a+\frac{b\,k^2}{n\,(n+1)}-\frac{b\,k}{n\,(n+1)}-\frac{a\,k^2}{n\,(n+1)}+\frac{a\,k}{n\,(n+1)}$$

Now we define our first quadrature, an adaptive trapezoidal's rule:

> *adtrap:=proc(a,b,n) sum(ank(a, b, n, k)*(f(left(a, b, j, k, n))+f(right(a, b, j, k, n)))*
/2, k=2..n)
> *end;*

adtrap :=

proc(a,b,n) sum(1/2* ank(a,b,n,k)*(f(left(a,b,j,k,n))+f(right(a,b,j,k,n))) ,k = 2 .. n)
end

Note that since the singularity of the function f is at $x = 0$, our quadrature "*adtrap*" described above starts evaluating at the second point, in other words, we avoid the singularity. Let's experiment our quadrature as follows with different number of divisions.

> *evalf(adtrap(0,1,500));evalf(adtrap(0,1,600));evalf(adtrap(0,1,630));*

1.995080118

1.995899693

1.996094852

In view of the data obtained above, we conjecture that the integral is convergent; of course we know that the improper integral of this function exists and is equal to 2. The next experiment we would like to see is what if we change our matrix ank, will we achieve better or worse convergence, and why? Let us use the formula $\sum_{k=1}^{n} k^2 = \frac{n(n+1)(2n+1)}{6}$ to form the matrix bnk as follows:

> *bnk:=proc(a,b,n,k) (6*(b-a)*(k**2)/(n*(n+1)*(2*n+1))) end;*

```
bnk := proc(a,b,n,k) 6*(b-a)*(k**2)/n/(n+1)/(2*n+1) end
```

Follow the procedure we defined the quadrature "*adtrap*", we define the second quadrature "*Adtrap*" by using the matrix "*bnk*".

```
> Right:=proc(a,b,j,k,n) a+sum(bnk(a,b,n,j), j=1..k) end;
```

```
Right := proc(a,b,j,k,n) a+sum(bnk(a,b,n,j),j = 1 .. k) end
```

```
> Left:=proc(a,b,j,k,n) a+sum(bnk(a,b,n,j), j=0..k-1) end;
```

```
Left := proc(a,b,j,k,n) a+sum(bnk(a,b,n,j),j = 0 .. k-1) end
```

```
> Adtrap:=proc(a,b,n) sum(bnk(a,b,n,k)*(f(Left(a,b,j,k,n))+f(Right(a,b,j,k,n)))/2, k=
2..n) end;
```

```
Adtrap :=
proc(a,b,n) sum(1/2* bnk(a,b,n,k)*(f(Left(a,b,j,k,n))+f(Right(a,b,j,k,n))) ,k = 2 .. n)
end
```

```
> evalf(Adtrap(0,1,80));evalf(Adtrap(0,1,90));evalf(Adtrap(0,1,100));
```

$$1.998705775$$

$$1.998939100$$

$$1.999111418$$

Clearly, by comparing the quadratures "*adtrap*" and "*Adtrap*", we see that the second quadrature incorporating the matrix "*bnk*" gives a better convergence and will use less number of points in evaluations. Why? This is because the choice of the matrix *bnk* is "compatible" with the behavior of the function f. Next we shall describe an adaptive Simpson's rule by using the matrix "*bnk*". First we need to define the midpoint of each subinterval.

```
> Mid:= proc(a,b,j,k,n) (Right(a,b,j,k,n)+Left(a,b,j,k,n))/2 end;
```

```
Mid := proc(a,b,j,k,n) 1/2*Right(a,b,j,k,n)+1/2*Left(a,b,j,k,n) end
```

Here is the adaptive Simpson's rule with the matrix "*bnk*".

> *Adsim:=proc(a,b,n) sum((bnk(a,b,n,k)/6)*(f(Left(a,b,j,k,n))+4*f(Mid(a,b,j,k,n))+f(Right(a,b,j,k,n))), k=2..n) end;*

Adsim := proc(a,b,n) sum(1/6*bnk(a,b,n,k)*(f(Left(a,b,j,k,n))+4* f(Mid(a,b,j,k,n))+ f(Right(a,b,j,k,n))),k = 2 .. n) end

Let's compare *Adsim* with *Adtrap* as follows:

> *evalf(Adsim(0,1,80));evalf(Adsim(0,1,90));evalf(Adsim(0,1,100));*

1.995311595

1.996066815

1.996639020

We observe that the quadrature "*Adtrap*" is better than that of "*Adsim*". Notice that for each fixed interval the errors for "*Adtrap*" and "*Adsim*" are

$$\sum_{k=1}^{n} \frac{-f''(\eta)a_{nk}^3}{12}$$

and

$$\sum_{k=1}^{n} \frac{-f^{(4)}(\eta)(a_{nk}/2)^5}{90}$$

respectively. Apparently, the behavior of f makes the quadrature "*Adtrap*" a better choice. Finally, we would like to modify the adaptive Trapezoidal's rule "*Adtrap*" to achieve a better result. We remark that the quadrature "*Adtrap*" ignores the tail completely since we wanted to avoid the singularity at $x = 0$. Now we want to add this tail, $a_{n1}f(x_0)$, where x_0 = the first right end point, and we predict we will obtain something better.

> *best:=proc(a,b,n) bnk(a,b,n,1)*f(Right(a,b,j,1,n))+sum(bnk(a,b,n,k)*(f(Left(a,b,j,k,n))+f(Right(a,b,j,k,n)))/2, k=2..n) end;*

best :=
proc(a,b,n) bnk(a,b,n,1)*f(Right(a,b,j,1,n))+sum(1/2* bnk(a,b,n,k)*(f(Left(a,b,j,k,n))+f(Right(a,b,j,k,n))),k = 2 .. n) end

> *evalf(best(0,1,80));evalf(best(0,1,90));evalf(best(0,1,100));*

2.001103919

2.000950944

2.000830584

Remarks: (1) The quadratures decribed in this paper could be lifted to two dimensions, see [2].
(2) Given a function, how to pick the best matrix for this function needs further investigations.
(3) Sometimes, Maple will not perform calculations for our quadratures if the number of divisions gets large. However, we could always write a program to run our quadratures such as using Turbo Pascal, Fortran, or etc. Computer algebra systems help us make conjectures which could not have been accomplished otherwise.

Conclusions: Computer algebra systems have been used in teaching calculus, linear algebra, differential equations, and other courses. Text books incorporated with CAS are popular items too. How to use CAS in making mathematical conjectures and forming new theorems beyond the first and second year's of undergraduate courses is an important issue as indicated in [5].

References:

[1] P.Y. Lee, *Lanzhou Lectures on Henstock integration*, World Scientic, Singapore, 1989.
[2] P.Y. Lee and W.-C. Yang, *Henstock integral and numerical integrations*, to be submitted.
[3] R.M. McLeod, *The Generalized Riemann Integral*, Mathematical Association of America, Washington, D.C., 1980.
[4] W.F. Pfeffer, *The Riemann approach to integration*, Cambridge, 1993.
[5] W.-C. Yang, *Post Calculus Analysis and Maple*, Conference Proceedings of the 6th ICTCM, Addison-Wesley.